The ODE’s that are used are:

1) sept2.ode, for microtubule polymerization (See Ref. Model for spatial microtubule oscillations, D. Sept, Phys Rev. E, 60, 838):

```
# full sept model
# play with C lower until oscillations go away
par c=50
par kp=9,kr=2,kc=.001,kn=.000005,ki=.2
tt=c-ta-td
n=kn*tt/(ki*td+kc)
  # n'=-kc*n+kn*tt-ki*n*td
ta'=kp*n*tt-kc*ta+kn*tt-ki*ta*td
td'=kc*ta-kr*td+ki*ta*td
@ total=200,dt=.05,meth=rk4,maxstor=5000
@ bound=1000
@ xp=ta,yp=td,xlo=0,ylo=0,xhi=20,yhi=4
aux tut=tt
aux nm=n
done
```

2) thermrat.ode, for Brownian rachets

```
# a simple thermal ratchet
wiener w
par a=.8,sig=.05,alpha=.1,beta=.1
# piecewise linear potential with slope
# 1 from 0 to a and slope -a/(1-a) from a to 1
# f = -V'
f(x)=if(x<a)then(-1)else(a/(1-a))
x'=z*f(mod(x,1))+sig*w
# z is two states
markov z 2
{0} {alpha}
{beta} {0}
@ meth=euler,dt=.1,total=2000,njmp=10
@ xhi=2000,yhi=8,ylo=-8
done
```
3) gib.ode, which uses Gillespie algorithm

```plaintext
# gillesp_bruss.ode
# gillespie algorithm for brusselator
#
# x1  ->  y1  (c1)
# x2+y1  ->  y2+Z  (c2)
# 2 y1 + y2  ->  3 y1  (c3)
# y1  ->  Z2  (c4)
par c1x1=5000,c2x2=50,c3=.00005,c4=5
init y1=1000,y2=2000
# compute the cumulative reactions
p1=c1x1
p2=p1+c2x2*y1
p3=p2+c3*y1*y2*(y1-1)/2
p4=p3+c4*y1
# choose random#
s2=ran(1)*p4
z1=(s2<p1)
z2=(s2<p2)&(s2>=p1)
z3=(s2<p3)&(s2>=p2)
z4=s2>p3
# time for next reaction
tr'=tr-log(ran(1))/p4
y1'=max(0,y1+z1-z2+z3-z4)
y2'=max(0,y2+z2-z3)
@ bound=100000000,meth=discrete,total=1000000,njmp=1000
@ xp=y1,yp=y2
@ xlo=0,ylo=0,xhi=10000,yhi=10000
done
```
The results of these ODE’s:

1) sept2.ode, for microtubule polymerization

i) Commends: Initialconds, Go

![Figure 1. $T_D$ versus $T_A$](image1)

ii) Commends: $X_i$ vs $t$, $T_D$, Graphics Stuff, Freeze, Freeze, Color 1, $X_i$ vs $t$, $T_A$

![Figure 2. $T_A$ and $T_D$ versus time, note the damped oscillations](image2)

iii) Lower C to 10 and see that oscillations are gone:
Commends: param, c:10, ok, go

![Figure 3.](image3)
Figure 3. $T_D$ versus time (note that oscillations are gone)

2) thermrat.ode, for Brownian rachets

i) $a=0.8$

Commends: Numerics, stochast, compute, range over X, 20 steps, start 0, end 0

Figure 4. 20 trajectories for X

Commends: stochast, mean, esc, erase, restore

Figure 5. Average of 20 trajectories for X
ii) \( a=0.2 \)

Commends: param, \( a=0.2 \), Numerics, stochast, compute, range over \( X \), 20 steps, start 0, end 0

Figure 6. 20 trajectories for \( X \)

Commends: stochast, mean, esc, erase, restore

Figure 7. Average of 20 trajectories for \( X \)
iii) \( a=0.5 \)

Commends: param, \( a=0.5 \), Numerics, stochast, compute, range over \( X \), 20 steps, start 0, end 0

Figure 8. 20 trajectories for \( X \)

Commends: stochast, mean, esc, erase, restore

Figure 9. Average of 20 trajectories for \( X \)
3) `gib.ode`, which uses Gillespie algorithm

![Figure 10. Y2 versus t](image)

Commends: param c1x1=500, c2x2=50, ok, close, ICs, Y1=100, Y2=200, ok, close

![Figure 11. Y2 versus t. Note the noise](image)